Cálculo 3 - 2024.2

P1 (primeira prova)

Eduardo Ochs - RCN/PURO/UFF
http://anggtwu.net/2024.2-C3.html

Links

http://anggtwu.net/e/maxima.e.html#2024.2-C3-P1-Q1
http://anggtwu.net/e/maxima.e.html#2024.2-C3-P1-Q2
(find-es "maxima" "2024.2-C3-P1-Q1")
(find-es "maxima" "2024.2-C3-P1-Q2")

Questão 1

(Total: 3.5 pts)

O diagrama de numerozinhos da última folha da prova corresponde a uma superfície $z = F(x, y)$ que tem 6 faces. Também é possível interpretá-lo como uma superfície com 7 ou mais faces, mas vamos considerar que a superfície com só 6 faces é que é a correta.

a) **(0.5 pts)** Mostre como dividir o plano em 6 polígonos que são as projeções destas faces no plano do papel.

b) **(0.5 pts)** Chame estas faces de face N ("norte"), S ("sul"), W ("oeste"), C ("centro"), E ("leste") e NE ("nordeste"), e chame as equações dos planos delas de $F_N(x, y)$, $F_S(x, y)$, $F_W(x, y)$, $F_C(x, y)$, $F_E(x, y)$, e $F_{NE}(x, y)$. Dê as equações destes planos.

c) **(0.5 pts)** Sejam:

$$\begin{aligned} P_C &= \{\, (x, y, z) \in \mathbb{R}^3 \mid z = F_C(x, y) \,\}, \\ P_E &= \{\, (x, y, z) \in \mathbb{R}^3 \mid z = F_E(x, y) \,\}, \\ r &= P_C \cap P_E. \end{aligned}$$

Represente a reta r graficamente como numerozinhos.

d) **(0.5 pts)** Dê uma parametrização para a reta do item anterior. Use notação de conjuntos.

e) **(0.5 pts)** Seja

$$A = \{0, 1, \ldots, 9\} \times \{0, 1, \ldots, 11\};$$

note que os numerozinhos do diagrama de numerozinhos estão todos sobre pontos de A. Para cada ponto $(x, y) \in A$ represente graficamente $(x, y) + \frac{1}{3}\vec{\nabla} F(x, y)$.

Obs: quando $\vec{\nabla} F(x, y) = 0$ desenhe uma bolinha preta sobre o ponto (x, y), e quando $\vec{\nabla} F(x, y)$ não existir faça um '×' sobre o numerozinho que está no ponto (x, y).

f) **(1.0 pts)** Sejam

$$\begin{aligned} Q(t) &= (0, 2) + t\overrightarrow{(1, 1)}, \\ (x(t), y(t)) &= Q(t), \\ h(t) &= F(x(t), y(t)). \end{aligned}$$

Faça o gráfico da função $h(t)$. Considere que o domínio dela é o intervalo $[0, 9]$.

Algumas definições

Em Cálculo 1 e Cálculo 2 você viu que se $f(x)$ é uma função de $\mathbb{R}$ em $\mathbb{R}$ então a aproximação de Taylor de ordem 2 pra $f(x)$ no ponto x_0 é:

$$\begin{array}{rcl}(T_{2,x_0}f)(x) &=& f(x_0)\\ &+& f'(x_0)\Delta x\\ &+& \frac{f''(x_0)}{2}\Delta x^2\end{array}$$

A "versão Cálculo 3" disto é a fórmula abaixo. Se $F(x,y)$ é uma função de $\mathbb{R}^2$ em $\mathbb{R}$ então a aproximação de Taylor de ordem 2 pra $F(x,y)$ no ponto (x_0,y_0) é:

$$\begin{array}{rcl}(T_{2,(x_0,y_0)}F)(x) &=& F(x_0,y_0)\\ &+& F_x(x_0,y_0)\Delta x + F_y(x_0,y_0)\Delta y\\ &+& \frac{F_{xx}(x_0,y_0)}{2}\Delta x^2 + F_{xy}(x_0,y_0)\Delta x\Delta y + \frac{F_{yy}(x_0,y_0)}{2}\Delta y^2\end{array}$$

e a gente diz que as derivadas até ordem 2 da função F são as funções $(F, F_x, F_y, F_{xx}, F_{xy}, F_{yy})$. Eu costumo organizar elas numa matriz:

$$D_2F = \begin{pmatrix} F & & \\ F_x & F_y & \\ F_{xx} & F_{xy} & F_{yy}\end{pmatrix}$$

$$(D_2F)(x_0,y_0) = \begin{pmatrix} F(x_0,y_0) & & \\ F_x(x_0,y_0) & F_y(x_0,y_0) & \\ F_{xx}(x_0,y_0) & F_{xy}(x_0,y_0) & F_{yy}(x_0,y_0)\end{pmatrix}$$

Questão 2

(Total: 6.5 pts)
Sejam

$$\begin{aligned} F(x,y) &= xy(6-2x-y), \\ P_1 &= (0,6), \\ P_2 &= (1,2), \\ P_3 &= (3,0), \\ P_4 &= (0,0). \end{aligned}$$

a) **(0.5 pts)** Calcule D_2F.

b) **(0.5 pts)** Calcule D_2F nos pontos P_1, P_2, P_3, e P_4.

c) **(1.0 pts)** Calcule $T_{2,(x_0,y_0)}F$ nos pontos P_1, P_2, P_3, e P_4.

d) **(0.5 pts)** Os pontos P_1, P_2, P_3 e P_4 são pontos críticos da função F? Quais deles são máximos locais? Quais são mínimos locais? Quais são pontos de sela? Use o gradiente e o determinante $\left|\begin{smallmatrix} F_{xx} & F_{xy} \\ F_{yx} & F_{yy} \end{smallmatrix}\right|$ pra descobrir tudo isso.

Lembre que $P_2 = (1,2)$.
Seja $G(x,y) = (T_{2,(1,2)}F)(x,y)$.
Seja $B = \{0,...,3\} \times \{0,...,6\}$
e $C = \{\, (x,y) \in B \mid y \leq 6 - 2x \,\}$.

e) **(0.5 pts)** Calcule o diagrama de numerozinhos da função F nos pontos de C.

f) **(1.0 pts)** Calcule o diagrama de numerozinhos da função G nos pontos de C.

g) **(2.5 pts)** Use o diagrama de numerozinhos da F que você calculou no item (e) e os gradientes da F nos pontos de C – que você ainda não calculou, e vai ter que calcular agora – pra fazer um desenho bem caprichado das curvas de nível da F dentro do triângulo cujos vértices são os pontos P_1, P_3 e P_4. Você vai precisar reduzir a escala dos vetores gradientes pra que eles não esbarrem uns nos outros – desenhe $F(x,y) + \frac{1}{10}\nabla F(x,y)$ para cada ponto de C.



Questão 1: gabarito (1a)



Questão 1: gabarito (1a, 1b)

```
(%i1) mkmatrix5(x,xs,y,ys,expr) ::=
        buildq([x,xs,y,ys,expr],
               apply('matrix,
                     makelist(makelist(expr,x,xs),y,ys)))$
(%i2) /* (1a: 0.5 pts) */
      /* (1b: 0.5 pts) */
      z_N  : 6$
(%i3) z_S  : 0$
(%i4) z_W  : y - 1;
(%o4)                              y − 1

(%i5) z_C  : y - x + 1;
(%o5)                            y − x + 1

(%i6) z_E  : -12 + 2*y;
(%o6)                             2 y − 12

(%i7) z_NE : -4 + y;
(%o7)                              y − 4

(%i8) z_MR : min(z_E, z_NE);              /* middle right */
(%o8)                       min (y − 4, 2 y − 12)

(%i9) z_M  : min(z_W, max(z_C, z_MR));  /* middle */
(%o9)       min (max (min (y − 4, 2 y − 12) , y − x + 1) , y − 1)

(%i10) z    : min(z_N, max(z_S, z_M))$
```

```
(%i11) mkmatrix5(x,seq(0,9), y,seqby(11,0,-1), [x,y]);
(%o11)
⎛ [0,11]  [1,11]  [2,11]  [3,11]  [4,11]  [5,11]  [6,11]  [7,11]  [8,11]  [9,11] ⎞
⎜ [0,10]  [1,10]  [2,10]  [3,10]  [4,10]  [5,10]  [6,10]  [7,10]  [8,10]  [9,10] ⎟
⎜  [0,9]   [1,9]   [2,9]   [3,9]   [4,9]   [5,9]   [6,9]   [7,9]   [8,9]   [9,9]  ⎟
⎜  [0,8]   [1,8]   [2,8]   [3,8]   [4,8]   [5,8]   [6,8]   [7,8]   [8,8]   [9,8]  ⎟
⎜  [0,7]   [1,7]   [2,7]   [3,7]   [4,7]   [5,7]   [6,7]   [7,7]   [8,7]   [9,7]  ⎟
⎜  [0,6]   [1,6]   [2,6]   [3,6]   [4,6]   [5,6]   [6,6]   [7,6]   [8,6]   [9,6]  ⎟
⎜  [0,5]   [1,5]   [2,5]   [3,5]   [4,5]   [5,5]   [6,5]   [7,5]   [8,5]   [9,5]  ⎟
⎜  [0,4]   [1,4]   [2,4]   [3,4]   [4,4]   [5,4]   [6,4]   [7,4]   [8,4]   [9,4]  ⎟
⎜  [0,3]   [1,3]   [2,3]   [3,3]   [4,3]   [5,3]   [6,3]   [7,3]   [8,3]   [9,3]  ⎟
⎜  [0,2]   [1,2]   [2,2]   [3,2]   [4,2]   [5,2]   [6,2]   [7,2]   [8,2]   [9,2]  ⎟
⎜  [0,1]   [1,1]   [2,1]   [3,1]   [4,1]   [5,1]   [6,1]   [7,1]   [8,1]   [9,1]  ⎟
⎝  [0,0]   [1,0]   [2,0]   [3,0]   [4,0]   [5,0]   [6,0]   [7,0]   [8,0]   [9,0]  ⎠
(%i12) mkmatrix5(x,seq(0,8), y,seqby(11,0,-1), ''z);
(%o12)
⎛ 6  6  6  6  6  6  6  6  6 ⎞
⎜ 6  6  6  6  6  6  6  6  6 ⎟
⎜ 6  6  6  6  6  5  5  5  5 ⎟
⎜ 6  6  6  6  5  4  4  4  4 ⎟
⎜ 6  6  6  5  4  3  2  2  2 ⎟
⎜ 5  5  5  4  3  2  1  0  0 ⎟
⎜ 4  4  4  3  2  1  0  0  0 ⎟
⎜ 3  3  3  2  1  0  0  0  0 ⎟
⎜ 2  2  2  1  0  0  0  0  0 ⎟
⎜ 1  1  1  0  0  0  0  0  0 ⎟
⎜ 0  0  0  0  0  0  0  0  0 ⎟
⎝ 0  0  0  0  0  0  0  0  0 ⎠
(%i13) /*
       plot3d (z,    [x,0,8],    [y,0,11]);
       */
```

Questão 1: gabarito (1c, 1d)

```
(%i13) /* (1c: 0.5 pts) */
                    [zr_=z_C, zr_=z_E];
```

(%o13)

$$[\mathrm{zr\_} = y - x + 1, \mathrm{zr\_} = 2\,y - 12]$$

```
(%i14)        solve([zr_=z_C, zr_=z_E], [y,zr_]);
```

(%o14)

$$[[y = 13 - x, \mathrm{zr\_} = 14 - 2\,x]]$$

```
(%i15) eqc : solve([zr_=z_C, zr_=z_E], [y,zr_])[1];
```

(%o15)

$$[y = 13 - x, \mathrm{zr\_} = 14 - 2\,x]$$

```
(%i16) define(yr_(x), subst(eqc, y));
```

(%o16)

$$\mathrm{yr\_}(x) := 13 - x$$

```
(%i17) define(zr_(x), subst(eqc, zr_));
```

(%o17)

$$\mathrm{zr\_}(x) := 14 - 2\,x$$

```
(%i18) xyzr(x)     := [x, yr_(x), zr_(x)];
```

(%o18)

$$\mathrm{xyzr}(x) := [x, \mathrm{yr\_}(x), \mathrm{zr\_}(x)]$$

```
(%i19) xyzr_top    : rhs(fundef(xyzr));
```

(%o19)

$$[x, \mathrm{yr\_}(x), \mathrm{zr\_}(x)]$$

```
(%i20) xyzr_lines : makelist(xyzr(x), x,2,9);
```

(%o20)

$$[[2, 11, 10], [3, 10, 8], [4, 9, 6], [5, 8, 4], [6, 7, 2], [7, 6, 0], [8, 5, -2], [9, 4, -4]]$$

```
(%i21) apply('matrix, append([xyzr_top], xyzr_lines));
```

(%o21)

$$\begin{pmatrix} x & \mathrm{yr\_}(x) & \mathrm{zr\_}(x) \\ 2 & 11 & 10 \\ 3 & 10 & 8 \\ 4 & 9 & 6 \\ 5 & 8 & 4 \\ 6 & 7 & 2 \\ 7 & 6 & 0 \\ 8 & 5 & -2 \\ 9 & 4 & -4 \end{pmatrix}$$

```
(%i22)  /* (1d: 0.5 pts) */
        [x, yr_(x), zr_(x)];
```

(%o22)

$$[x, 13 - x, 14 - 2\,x]$$

Questão 1: gabarito (1e, 1f)

```
(%i23) /* (1e: 0.5 pts) */
       define(z(x,y), z);
```
(%o23)
$$z(x,y) := \min(6, \max(0, \min(\max(\min(y-4, 2\,y-12), y-x+1), y-1)))$$

```
(%i24) eps : 1/4;
```
(%o24)
$$\frac{1}{4}$$

```
(%i25) z_xr (x,y) := (z(x+eps,y)-z(x,y))/ eps;
```
(%o25)
$$z\_xr(x,y) := \frac{z(x+\text{eps},y)-z(x,y)}{\text{eps}}$$

```
(%i26) z_xl (x,y) := (z(x-eps,y)-z(x,y))/-eps;
```
(%o26)
$$z\_xl(x,y) := \frac{z(x-\text{eps},y)-z(x,y)}{-\text{eps}}$$

```
(%i27) z_yu (x,y) := (z(x,y+eps)-z(x,y))/ eps;
```
(%o27)
$$z\_yu(x,y) := \frac{z(x,y+\text{eps})-z(x,y)}{\text{eps}}$$

```
(%i28) z_yd (x,y) := (z(x,y-eps)-z(x,y))/-eps;
```
(%o28)
$$z\_yd(x,y) := \frac{z(x,y-\text{eps})-z(x,y)}{-\text{eps}}$$

```
(%i29) gradz(x,y) := if (z_xr(x,y) = z_xl(x,y)) and
                       (z_yu(x,y) = z_yd(x,y))
                     then [z_xr(x,y), z_yu(x,y)]
                     else "X"$
```

```
(%i30) mkmatrix5(x,seq(0,8), y,seqby(11,0,-1), gradz(x,y));
```
(%o30)
$$\begin{pmatrix}
[0,0] & [0,0] & [0,0] & [0,0] & [0,0] & [0,0] & [0,0] & [0,0] & [0,0] \\
[0,0] & [0,0] & [0,0] & [0,0] & [0,0] & X & X & X & X \\
[0,0] & [0,0] & [0,0] & [0,0] & X & X & [0,1] & [0,1] & [0,1] \\
[0,0] & [0,0] & [0,0] & X & [-1,1] & X & X & X & X \\
X & X & X & [-1,1] & [-1,1] & [-1,1] & X & [0,2] & [0,2] \\
[0,1] & [0,1] & X & [-1,1] & [-1,1] & [-1,1] & [-1,1] & X & X \\
[0,1] & [0,1] & X & [-1,1] & [-1,1] & [-1,1] & X & [0,0] & [0,0] \\
[0,1] & [0,1] & X & [-1,1] & [-1,1] & X & [0,0] & [0,0] & [0,0] \\
[0,1] & [0,1] & X & [-1,1] & X & [0,0] & [0,0] & [0,0] & [0,0] \\
[0,1] & [0,1] & X & X & [0,0] & [0,0] & [0,0] & [0,0] & [0,0] \\
X & X & X & [0,0] & [0,0] & [0,0] & [0,0] & [0,0] & [0,0] \\
[0,0] & [0,0] & [0,0] & [0,0] & [0,0] & [0,0] & [0,0] & [0,0] & [0,0]
\end{pmatrix}$$

```
(%i31)
       /* (1f: 1.0 pts) */
       [xmin,xmax, ymin,ymax] : [0,9, 0,7];
```
(%o31)
$$[0, 9, 0, 7]$$

```
(%i32) Q(t) := [0,2] + t*[1,1];
```
(%o32)
$$Q(t) := [0,2] + t\,[1,1]$$

```
(%i33) define(xQ(t), Q(t)[1]);
```
(%o33)
$$\text{xQ}(t) := t$$

```
(%i34) define(yQ(t), Q(t)[2]);
```
(%o34)
$$\text{yQ}(t) := t + 2$$

```
(%i35)                          [x=xQ(t),x=yQ(t)];
```
(%o35)
$$[x = t, x = t + 2]$$

Questão 1: gabarito (1f)

```
(%i136) define(h(t), at(z, [x=xQ(t),y=yQ(t)]));
(%o36)
```

$$h\left(t\right):=\min\left(6,\max\left(0,\min\left(\max\left(3,\min\left(t-2,2\left(t+2\right)-12\right)\right),t+1\right)\right)\right)$$

```
(%i137) myqdrawp(xyrange(), myex1(h(x), lc(red)));
(%o37)
```




```
(%i138)
```

Questão 2: gabarito

```
(%i1) gradef(W  (x,y), W_x (x,y), W_y (x,y))$
(%i2) gradef(W_x(x,y), W_xx(x,y), W_xy(x,y))$
(%i3) gradef(W_y(x,y), W_xy(x,y), W_yy(x,y))$
(%i4) dd(F)       := [F,
                      diff(F,x),   diff(F,y),
                      diff(F,x,2), diff(F,x,1,y,1), diff(F,y,2)]$
(%i5) aa(o,x0y0)  := at(o, [x=x0y0[1], y=x0y0[2]])$
(%i6) mm(abcdef)  := block([a,b,c,d,e,f],
                           [a,b,c,d,e,f]:abcdef,
                           [a,b+Dx,c+Dy,d+Dx^2/2,e+Dx+Dy,f+Dy^2/2])$
(%i7) ss(abcdef)  := block([a,b,c,d,e,f],
                           [a,b,c,d,e,f]:abcdef,
                           a+b+c+d+e+f)$
(%i8) toM(abcdef) := block([a,b,c,d,e,f],
                           [a,b,c,d,e,f]:abcdef,
                           matrix([a,"",""], [b,c,""], [d,e,f]))$
(%i9) D2       (F) := toM(dd(F))$
(%i10) D2at(x0y0,F) := toM(aa(dd(F),x0y0))$
(%i11) T2M (x0y0,F) := toM(mm(aa(dd(F),x0y0)))$
(%i12) T2  (x0y0,F) :=  ss(mm(aa(dd(F),x0y0)))$
(%i13)
       /* Alguns testes: */
                dd(W(x,y));
(%o13)
 [W(x,y), W_x(x,y), W_y(x,y), W_xx(x,y), W_xy(x,y), W_yy(x,y)]

(%i14)       toM(dd(W(x,y)));
(%o14)
                          ⎛ W(x,y)     ⎞
                          ⎜ W_x(x,y)   W_y(x,y)   ⎟
                          ⎝ W_xx(x,y)  W_xy(x,y)  W_yy(x,y) ⎠
(%i15)       toM([1,2,3,4,5,6]);
(%o15)
                                ⎛ 1 ⎞
                                ⎜ 2  3 ⎟
                                ⎝ 4  5  6 ⎠
(%i16)        aa(dd(W(x,y)),[3,4]);
(%o16)
 [W(3,4), W_x(3,4), W_y(3,4), W_xx(3,4), W_xy(3,4), W_yy(3,4)]

(%i17)    toM(aa(dd(W(x,y)),[3,4]));
(%o17)
                          ⎛ W(3,4)     ⎞
                          ⎜ W_x(3,4)   W_y(3,4)   ⎟
                          ⎝ W_xx(3,4)  W_xy(3,4)  W_yy(3,4) ⎠
(%i18)     mm(aa(dd(W(x,y)),[3,4]));
(%o18)
 [W(3,4), W_x(3,4) Dx, W_y(3,4) Dy, (W_xx(3,4) Dx^2)/2, W_xy(3,4) Dx Dy, (W_yy(3,4) Dy^2)/2]

(%i19) toM(mm(aa(dd(W(x,y)),[3,4])));
(%o19)
                          ⎛ W(3,4)     ⎞
                          ⎜ W_x(3,4) Dx    W_y(3,4) Dy   ⎟
                          ⎝ (W_xx(3,4) Dx^2)/2  W_xy(3,4) Dx Dy  (W_yy(3,4) Dy^2)/2 ⎠
(%i20)  ss(mm(aa(dd(W(x,y)),[3,4])));
(%o20)
 (W_yy(3,4) Dy^2)/2 + W_xy(3,4) Dx Dy + W_y(3,4) Dy + (W_xx(3,4) Dx^2)/2 + W_x(3,4) Dx + W(3,4)

(%i21)
       D2        (W(x,y));
(%o21)
                          ⎛ W(x,y)     ⎞
                          ⎜ W_x(x,y)   W_y(x,y)   ⎟
                          ⎝ W_xx(x,y)  W_xy(x,y)  W_yy(x,y) ⎠
(%i22) D2at([3,4],W(x,y));
(%o22)
                          ⎛ W(3,4)     ⎞
                          ⎜ W_x(3,4)   W_y(3,4)   ⎟
                          ⎝ W_xx(3,4)  W_xy(3,4)  W_yy(3,4) ⎠
(%i23) T2M ([3,4],W(x,y));
(%o23)
                          ⎛ W(3,4)     ⎞
                          ⎜ W_x(3,4) Dx    W_y(3,4) Dy   ⎟
                          ⎝ (W_xx(3,4) Dx^2)/2  W_xy(3,4) Dx Dy  (W_yy(3,4) Dy^2)/2 ⎠
(%i24) T2  ([3,4],W(x,y));
(%o24)
 (W_yy(3,4) Dy^2)/2 + W_xy(3,4) Dx Dy + W_y(3,4) Dy + (W_xx(3,4) Dx^2)/2 + W_x(3,4) Dx + W(3,4)
```

Questão 2: gabarito (2a, 2b, 2c)

(%i125) F : x+y*(6 - 2*x - y);
(%o125)
$$x\,(-y-2\,x+6)\,y$$

(%i126) F : expand(F);
(%o126)
$$-\left(x\,y^2\right)-2\,x^2\,y+6\,x\,y$$

(%i127) P1 : [0,6]$
(%i128) P2 : [1,2]$
(%i129) P3 : [3,0]$
(%i130) P4 : [0,0]$
(%i131) /* (2a: 0.5 pts) */
D2(W(x,y));
(%o131)
$$\begin{pmatrix} W\,(x,y) & \\ \mathrm{W\_x}\,(x,y) & \mathrm{W\_y}\,(x,y) \\ \mathrm{W\_xx}\,(x,y) & \mathrm{W\_xy}\,(x,y) \quad \mathrm{W\_yy}\,(x,y) \end{pmatrix}$$

(%i132) F;
(%o132)
$$-\left(x\,y^2\right)-2\,x^2\,y+6\,x\,y$$

(%i133) D2F : D2(F);
(%o133)
$$\begin{pmatrix} -\left(x\,y^2\right)-2\,x^2\,y+6\,x\,y & & \\ -y^2-4\,x\,y+6\,y & -\left(2\,x\,y\right)-2\,x^2+6\,x & \\ -\left(4\,y\right) & -\left(2\,y\right)-4\,x+6 & -\left(2\,x\right) \end{pmatrix}$$

(%i134) /* (2b: 0.5 pts) */
[P1, D2F, D2FP1:D2at(P1,F)];
(%o134)
$$\left[[0,6],\begin{pmatrix} -\left(x\,y^2\right)-2\,x^2\,y+6\,x\,y & & \\ -y^2-4\,x\,y+6\,y & -\left(2\,x\,y\right)-2\,x^2+6\,x & \\ -\left(4\,y\right) & -\left(2\,y\right)-4\,x+6 & -\left(2\,x\right) \end{pmatrix},\begin{pmatrix} 0 & & \\ 0 & 0 & \\ -24 & -6 & 0 \end{pmatrix}\right]$$

(%i135) [P2, D2F, D2FP2:D2at(P2,F)];
(%o135)
$$\left[[1,2],\begin{pmatrix} -\left(x\,y^2\right)-2\,x^2\,y+6\,x\,y & & \\ -y^2-4\,x\,y+6\,y & -\left(2\,x\,y\right)-2\,x^2+6\,x & \\ -\left(4\,y\right) & -\left(2\,y\right)-4\,x+6 & -\left(2\,x\right) \end{pmatrix},\begin{pmatrix} 4 & & \\ 0 & 0 & \\ -8 & -2 & -2 \end{pmatrix}\right]$$

(%i136) [P3, D2F, D2FP3:D2at(P3,F)];
(%o136)
$$\left[[3,0],\begin{pmatrix} -\left(x\,y^2\right)-2\,x^2\,y+6\,x\,y & & \\ -y^2-4\,x\,y+6\,y & -\left(2\,x\,y\right)-2\,x^2+6\,x & \\ -\left(4\,y\right) & -\left(2\,y\right)-4\,x+6 & -\left(2\,x\right) \end{pmatrix},\begin{pmatrix} 0 & & \\ 0 & 0 & \\ 0 & -6 & -6 \end{pmatrix}\right]$$

(%i137) [P4, D2F, D2FP4:D2at(P4,F)];
(%o137)
$$\left[[0,0],\begin{pmatrix} -\left(x\,y^2\right)-2\,x^2\,y+6\,x\,y & & \\ -y^2-4\,x\,y+6\,y & -\left(2\,x\,y\right)-2\,x^2+6\,x & \\ -\left(4\,y\right) & -\left(2\,y\right)-4\,x+6 & -\left(2\,x\right) \end{pmatrix},\begin{pmatrix} 0 & & \\ 0 & 0 & \\ 0 & 6 & 0 \end{pmatrix}\right]$$

(%i138) /* (2c: 1.0 pts) */
TM2([x0,y0],W(x,y));
(%o138)
$$\mathrm{TM2}\left([\mathrm{x0},\mathrm{y0}],W\,(x,y)\right)$$

(%i139) [P1, D2FP1, TM2(P1,F), T2(P1,F)];
(%o139)
$$\left[[0,6],\begin{pmatrix} 0 & & \\ 0 & 0 & \\ -24 & -6 & 0 \end{pmatrix},\begin{pmatrix} 0 & & \\ 0 & 0 & \\ -\left(12\,\mathrm{Dx}^2\right) & -\left(6\,\mathrm{Dx}\,\mathrm{Dy}\right) & 0 \end{pmatrix},-\left(6\,\mathrm{Dx}\,\mathrm{Dy}\right)-12\,\mathrm{Dx}^2\right]$$

(%i140) [P2, D2FP2, TM2(P2,F), T2(P2,F)];
(%o140)
$$\left[[1,2],\begin{pmatrix} 4 & & \\ 0 & 0 & \\ -8 & -2 & -2 \end{pmatrix},\begin{pmatrix} 4 & & \\ 0 & 0 & \\ -\left(4\,\mathrm{Dx}^2\right) & -\left(2\,\mathrm{Dx}\,\mathrm{Dy}\right) & -\mathrm{Dy}^2 \end{pmatrix},-\mathrm{Dy}^2-2\,\mathrm{Dx}\,\mathrm{Dy}-4\,\mathrm{Dx}^2+4\right]$$

(%i141) [P3, D2FP3, TM2(P3,F), T2(P3,F)];
(%o141)
$$\left[[3,0],\begin{pmatrix} 0 & & \\ 0 & 0 & \\ 0 & -6 & -6 \end{pmatrix},\begin{pmatrix} 0 & & \\ 0 & 0 & \\ 0 & -\left(6\,\mathrm{Dx}\,\mathrm{Dy}\right) & -\left(3\,\mathrm{Dy}^2\right) \end{pmatrix},-\left(3\,\mathrm{Dy}^2\right)-6\,\mathrm{Dx}\,\mathrm{Dy}\right]$$

(%i142) [P4, D2FP4, TM2(P4,F), T2(P4,F)];
(%o142)
$$\left[[0,0],\begin{pmatrix} 0 & & \\ 0 & 0 & \\ 0 & 6 & 0 \end{pmatrix},\begin{pmatrix} 0 & & \\ 0 & 0 & \\ 0 & 6\,\mathrm{Dx}\,\mathrm{Dy} & 0 \end{pmatrix},6\,\mathrm{Dx}\,\mathrm{Dy}\right]$$

Questão 2: gabarito (2d)

```
(%i143) /* (2d: 0.5 pts) */
        grad(F) := [diff(F,x),diff(F,y)]$
(%i144)    H(F) := hessian(F, [x,y])$
(%i145) detH(F) := determinant(H(F))$
(%i146) crit(F) := [F, grad(F), H(F), detH(F)]$
(%i147) crit(F) := matrix([F, grad(F)], [H(F), detH(F)])$
(%i148)    crit(W(x,y));
(%i148)
```

$$\begin{pmatrix} W(x,y) & [W\_x(x,y), W\_y(x,y)] \\ \begin{pmatrix} W\_xx(x,y) & W\_xy(x,y) \\ W\_xy(x,y) & W\_yy(x,y) \end{pmatrix} & W\_xx(x,y)\, W\_yy(x,y) - W\_xy(x,y)^2 \end{pmatrix}$$

```
(%i149) aa(crit(F), P1);
(%i149)
```

$$\begin{pmatrix} 0 & [0,0] \\ \begin{pmatrix} -24 & -6 \\ -6 & 0 \end{pmatrix} & -36 \end{pmatrix}$$

```
(%i150) aa(crit(F), P2);
(%i150)
```

$$\begin{pmatrix} 4 & [0,0] \\ \begin{pmatrix} -8 & -2 \\ -2 & -2 \end{pmatrix} & 12 \end{pmatrix}$$

```
(%i151) aa(crit(F), P3);
(%i151)
```

$$\begin{pmatrix} 0 & [0,0] \\ \begin{pmatrix} 0 & -6 \\ -6 & -6 \end{pmatrix} & -36 \end{pmatrix}$$

```
(%i152) aa(crit(F), P4);
(%i152)
```

$$\begin{pmatrix} 0 & [0,0] \\ \begin{pmatrix} 0 & 6 \\ 6 & 0 \end{pmatrix} & -36 \end{pmatrix}$$

```
(%i153) /* definicao da funcao G */
        P2;
(%i153)
```

$$[1,2]$$

```
(%i154) T2(P2,F);
(%i154)
```

$$-\mathrm{Dy}^2 - 2\,\mathrm{Dx}\,\mathrm{Dy} - 4\,\mathrm{Dx}^2 + 4$$

```
(%i155) G__ :          T2(P2,F);
(%i155)
```

$$-\mathrm{Dy}^2 - 2\,\mathrm{Dx}\,\mathrm{Dy} - 4\,\mathrm{Dx}^2 + 4$$

```
(%i156) G_  : subst([Dx=x-1,Dy=y-2], T2(P2,F));
(%i156)
```

$$-(2\,(x-1)\,(y-2)) - (y-2)^2 - 4\,(x-1)^2 + 4$$

```
(%i157) G   : expand(G_);
(%i157)
```

$$-y^2 - 2\,x\,y + 6\,y - 4\,x^2 + 12\,x - 8$$

```
(%i158) /* definicao dos conjuntos B e C */
        inC_(x,y)     : y <= 6 - 2*x$
(%i159) inC (x,y,o) := if inC_(x,y) then o else ""$
(%i160) numsB(expr) :=
          apply(matrix,
                makelist(makelist(ev(expr), x,0,3),
                         y, seqby(6,0,-1)))$
(%i161) numsC(expr) :=
          apply(matrix,
                makelist(makelist(inC(x,y,ev(expr)), x,0,3),
                         y, seqby(6,0,-1)))$
(%i162) numsB([x,y]);
(%i162)
```

$$\begin{pmatrix} [0,6] & [1,6] & [2,6] & [3,6] \\ [0,5] & [1,5] & [2,5] & [3,5] \\ [0,4] & [1,4] & [2,4] & [3,4] \\ [0,3] & [1,3] & [2,3] & [3,3] \\ [0,2] & [1,2] & [2,2] & [3,2] \\ [0,1] & [1,1] & [2,1] & [3,1] \\ [0,0] & [1,0] & [2,0] & [3,0] \end{pmatrix}$$

```
(%i163) numsC([x,y]);
(%i163)
```

$$\begin{pmatrix} [0,6] & & & \\ [0,5] & & & \\ [0,4] & [1,4] & & \\ [0,3] & [1,3] & & \\ [0,2] & [1,2] & [2,2] & \\ [0,1] & [1,1] & [2,1] & \\ [0,0] & [1,0] & [2,0] & [3,0] \end{pmatrix}$$

Questão 2: gabarito (2e, 2f)

```
(%i64) /* (2e: 0.5 pts) */
       F;
```

(%o64)
$$-\left(x y^2\right)-2 x^2 y+6 x y$$

```
(%i65) [numsC(   x  *y),
        numsC(  x^2*y),
        numsC(   x  *y^2)];
```

(%o65)
$$\left[\begin{pmatrix}0\\0\\0&4\\0&3\\0&2&4\\0&1&2\\0&0&0&0\end{pmatrix},\begin{pmatrix}0\\0\\0&4\\0&3\\0&2&8\\0&1&4\\0&0&0&0\end{pmatrix},\begin{pmatrix}0\\0\\0&16\\0&9\\0&4&8\\0&1&2\\0&0&0&0\end{pmatrix}\right]$$

```
(%i66) [numsC( 6*x  *y),
        numsC(-2*x^2*y),
        numsC(  -x  *y^2),
        numsC(F)];
```

(%o66)
$$\left[\begin{pmatrix}0\\0\\0&24\\0&18\\0&12&24\\0&6&12\\0&0&0&0\end{pmatrix},\begin{pmatrix}0\\0\\0&-8\\0&-6\\0&-4&-16\\0&-2&-8\\0&0&0&0\end{pmatrix},\begin{pmatrix}0\\0\\0&-16\\0&-9\\0&-4&-8\\0&-1&-2\\0&0&0&0\end{pmatrix},\begin{pmatrix}0\\0\\0&0\\0&3\\0&4&0\\0&3&2\\0&0&0&0\end{pmatrix}\right]$$

```
(%i67) /* (2f: 1.0 pts) */
       G_;
```

(%o67)
$$-\mathrm{Dy}^2-2\,\mathrm{Dx}\,\mathrm{Dy}-4\,\mathrm{Dx}^2+4$$

```
(%i68) G_;
```

(%o68)
$$-\left(2\left(x-1\right)\left(y-2\right)\right)-\left(y-2\right)^2-4\left(x-1\right)^2+4$$

```
(%i69) Dx : x-1$
(%i70) Dy : y-2$
(%i71) [numsC(Dx^2),
        numsC(Dx*Dy),
        numsC(Dy^2)];
```

(%o71)
$$\left[\begin{pmatrix}1\\1\\1&0\\1&0\\1&0&1\\1&0&1\\1&0&1&4\end{pmatrix},\begin{pmatrix}-4\\-3\\-2&0\\-1&0\\0&0&0\\1&0&-1\\2&0&-2&-4\end{pmatrix},\begin{pmatrix}16\\9\\4&4\\1&1\\0&0&0\\1&1&1\\4&4&4&4\end{pmatrix}\right]$$

```
(%i72) [numsC(4),
        numsC(-4*Dx^2),
        numsC(-2*Dx*Dy),
        numsC(  -Dy^2)];
```

(%o72)
$$\left[\begin{pmatrix}4\\4\\4&4\\4&4\\4&4&4\\4&4&4\\4&4&4&4\end{pmatrix},\begin{pmatrix}-4\\-4\\-4&0\\-4&0\\-4&0&-4\\-4&0&-4\\-4&0&-4&-16\end{pmatrix},\begin{pmatrix}8\\6\\4&0\\2&0\\0&0&0\\-2&0&2\\-4&0&4&8\end{pmatrix},\begin{pmatrix}-16\\-9\\-4&-4\\-1&-1\\0&0&0\\-1&-1&-1\\-4&-4&-4&-4\end{pmatrix}\right]$$

```
(%i73) numsC(G);
```

(%o73)
$$\begin{pmatrix}-8\\-3\\0&0\\1&3\\0&4&0\\-3&3&1\\-8&0&0&-8\end{pmatrix}$$

Questão 2: gabarito (2g)

```
(%i74) /* (2g: 2.5 pts) */
       grad(F);
```

(%o74)
$$\left[-y^2-4\,x\,y+6\,y,-\left(2\,x\,y\right)-2\,x^2+6\,x\right]$$

```
(%i75) Fx : diff(F,x);
```

(%o75)
$$-y^2-4\,x\,y+6\,y$$

```
(%i76)  -4*x*y + 6*y - y^2;
```

(%o76)
$$-y^2-4\,x\,y+6\,y$$

```
(%i77) [numsC(-4*x*y),
        numsC(6*y),
        numsC(-y^2),
        numsC(Fx)];
```

(%o77)
$$\left[\begin{pmatrix}0\\0\\0&-8\\0&-6\\0&-4&-8\\0&-2&-4\\0&0&0&0\end{pmatrix},\begin{pmatrix}36\\30\\24&24\\18&18\\12&12&12\\6&6&6\\0&0&0&0\end{pmatrix},\begin{pmatrix}-36\\-25\\-16&-16\\-9&-9\\-4&-4&-4\\-1&-1&-1\\0&0&0&0\end{pmatrix},\begin{pmatrix}0\\5\\8&-8\\9&-3\\8&0&-8\\5&1&-3\\0&0&0&0\end{pmatrix}\right]$$

```
(%i78) Fy : diff(F,y);
```

(%o78)
$$-\left(2\,x\,y\right)-2\,x^2+6\,x$$

```
(%i79)  -2*x*y + 6*x - 2*x^2;
```

(%o79)
$$-\left(2\,x\,y\right)-2\,x^2+6\,x$$

```
(%i80) [numsC(-2*x*y),
        numsC(6*x),
        numsC(-2*x^2),
        numsC(Fy)];
```

(%o80)
$$\left[\begin{pmatrix}0\\0\\0&-8\\0&-6\\0&-4&-8\\0&-2&-4\\0&0&0&0\end{pmatrix},\begin{pmatrix}0\\0\\0&6\\0&6\\0&6&12\\0&6&12\\0&6&12&18\end{pmatrix},\begin{pmatrix}0\\0\\0&-2\\0&-2\\0&-2&-8\\0&-2&-8\\0&-2&-8&-18\end{pmatrix},\begin{pmatrix}0\\0\\0&-4\\0&-2\\0&0&-4\\0&2&0\\0&4&4&0\end{pmatrix}\right]$$

```
(%i81) numsC(grad(F));
```

(%o81)
$$\begin{pmatrix}[0,0]\\[5,0]\\[8,0]&[-8,-4]\\[9,0]&[-3,-2]\\[8,0]&[0,0]&[-8,-4]\\[5,0]&[1,2]&[-3,0]\\[0,0]&[0,4]&[0,4]&[0,0]\end{pmatrix}$$

Questão 2: curvas de nível da F e da G





Questão 2: gradientes e curvas de nível da F